

CAPITOLO 11

I NUMERI COMPLESSI

Il linguaggio Qbasic è di notevole aiuto nel calcolo e nella presentazione grafica dei numeri complessi. Questo capitolo è dedicato a tale argomento che normalmente non è trattato mediante P.C. Le soluzioni software che sono illustrate, anche se di semplice struttura, consentono di ottenere risultati molto interessanti in questo particolare settore della matematica.

11.1 La presentazione grafica di un numero complesso

La presentazione grafica di un numero complesso, di per sé cosa scontata, è la base per calcoli più complicati che senza l'ausilio del P.C. sarebbero gravosi da svolgere.

Iniziamo pertanto in questo paragrafo con esercitazioni riguardanti tale tipo di presentazione:

Dati ad esempio tre numeri complessi

$$N1 = 3 + j 5$$

$$N2 = -7 + j 4$$

$$N3 = 6 - j 9$$

è possibile la loro rappresentazione grafica in un sistema di assi cartesiani a 4 quadranti mediante l'istruzione `LINE (xo , yo) - (X , Y)` in cui:

-nella prima parentesi sono indicate le coordinate del centro del reticolo, nel nostro caso $xo = 230$, $yo = 160$ (si veda il paragrafo 3.15).

-nella seconda parentesi sono indicate le variabili di posizione X e Y che, opportunamente riportate, identificano rispettivamente la parte reale e la parte immaginaria dei numeri complessi.

Il segmento che l'istruzione traccia sullo schermo è il vettore del generico punto N che rappresenta un numero complesso, la lunghezza del segmento è il modulo del vettore .

Ricordando quanto spiegato nel paragrafo 3.15 in merito alla costruzione generale dell'istruzione `PSET` per gli assi cartesiani a 4 quadranti abbiamo che le coordinate X e Y dell'istruzione `LINE` sono date da:

$$X = 230 + k \cdot x \qquad Y = 160 - k2 \cdot y$$

dove x è la parte reale di N e y la parte immaginaria.

Per la presentazione dei numeri complessi dobbiamo determinare i valori di k e $k2$ in base rispettivamente alle loro parti reali e immaginarie; per mantenere la stessa scala per entrambe le parti e dovendo scegliere per i due assi il valore superiore della parte più elevata, è comodo per la suddivisione del reticolo assumere il valore 10, si ha perciò:

$$k = 230 / 10 = 23 \qquad k2 = 160 / 10 = 16$$

e le istruzioni per la presentazione dei tre punti assumono la struttura:

$$\text{Per } N1 \quad \text{LINE (230,160) - (230 + 23 * 3 , 160 - 16 * 5)}$$

$$\text{Per } N2 \quad \text{LINE (230,160) - (230 + 23 * (-7) , 160 - 16 * 4)}$$

$$\text{Per } N3 \quad \text{LINE (230,160) - (230 + 23 * 6 , 160 - 16 * (-9))}$$

Nel reticolo del sistema cartesiano è opportuno marcare l'asse delle ordinate con **+j** nel semipiano superiore e **-j** nel semipiano inferiore.

Possiamo compilare il semplice programma di presentazione di N come segue:

```
' FORMAZIONE DEL SISTEMA DI ASSI CARTESIANI A 4 QUADRANTI CON INDICAZIONI
' SULL'ASSE DELLE ORDINATE DEI SIMBOLI +j -j
SCREEN 9
FOR x = 0 TO 460 STEP 23
FOR y = 0 TO 320 STEP 2 ' VERTICALE
PSET (x, y), 7
NEXT y
NEXT x
LOCATE 1, 30 : PRINT "+j" ' presentazione del simbolo +j
LOCATE 23, 30: PRINT "-j" ' presentazione del simbolo -j
FOR y = 0 TO 320 STEP 16
FOR x = 0 TO 460 STEP 3 ' ORIZZONTALE
PSET (x, y), 7
NEXT x
NEXT y
LINE ( 230, 0 )-( 230, 320 ) ' ASSE Y 4 quadranti
LINE ( 0, 160 )-( 460, 160 ) ' ASSE X 4 quadranti
LINE ( 230,160 ) - ( 230 + 23 * 3 , 160 - 16 * 5 ),14 ' vettore colore giallo
LINE ( 230,160 ) - ( 230 + 23 * ( -7 ) , 160 - 16 * 4 ) ' vettore colore bianco
LINE ( 230,160 ) - ( 230 + 23 * 6 , 160 - 16 * ( -9 ) ), 3 ' vettore colore turchese
```

F5

è presentato in figura 59 il sistema di assi cartesiani a 4 quadranti in cui compaiono:

- nel primo quadrante il vettore giallo che supporta il punto N1 all'incrocio del reticolo a cui corrispondono le coordinate $x = 3$ e $y = 5$
- nel secondo quadrante il vettore bianco che supporta N2 all'incrocio del reticolo a cui corrispondono le coordinate $x = -7$ e $y = 4$
- nel quarto quadrante il vettore turchese che supporta N3 all'incrocio del reticolo a cui corrispondono le coordinate $x = 6$ e $y = -9$.




Figura 59
Presentazione grafica
di 3 numeri complessi

11.2 Le operazioni sui numeri complessi

Prima di entrare nel dettaglio di questo paragrafo ricordiamo che un numero complesso $N = x + jy$ può essere espresso, sia in forma cartesiana, mediante i valori di x e y rispettivamente parte reale e parte immaginaria di N , sia in forma polare, mediante il modulo $|N|$ e l'argomento $\text{Arg}^\circ = \text{Arcotang}(y/x)$ di un **vettore** N . Pertanto, per completezza, i risultati delle operazioni forniranno sempre, sia la parte reale R e la parte immaginaria I , relativa alla rappresentazione cartesiana, sia il modulo $|N|$ e l'argomento Arg° relativi alla rappresentazione polare.

Come è noto le quattro operazioni aritmetiche di somma, sottrazione, moltiplicazione e divisione sono estensibili anche ai numeri complessi, le prime due si possono eseguire facilmente con carta e matita, le altre invece richiedono un certo lavoro di calcolo. Il Qbasic ci aiuta in questo lavoro dando modo di sviluppare rapidamente tanto il calcolo delle diverse operazioni, quanto la presentazione grafica del numero complesso risultante dal calcolo stesso.

Vediamo come scrivere in linguaggio Qbasic le quattro formule di cui abbiamo parlato:

Dato che il Qbasic non riconosce i numeri complessi questi devono essere scomposti nelle parti:

Reale = R

Immaginaria = I

Se $N = x + jy$ si ha:

$R = x$

$I = y$

soltanto in questo modo sono possibili le uguaglianze simboliche riportate nella pagina seguente.

Si deve osservare che l'uguaglianza simbolica relativa al calcolo dell'argomento, Arg° , del risultato di una operazione tra numeri complessi, è completata con l'addendo T che consente, al programma, di presentare i valori di Arg° sempre e soltanto come angoli positivi nel campo di variabilità compreso tra 0° e 360° .

operazione	simbologia ordinaria	simbologia Qbasic
somma	$(x_1 + jy_1) + (x_2 + jy_2)$	$R = x_1 + x_2 ; I = y_1 + y_2$
differenza	$(x_1 + jy_1) - (x_2 + jy_2)$	$R = x_1 - x_2 ; I = y_1 - y_2$
prodotto	$(x_1 + jy_1) \cdot (x_2 + jy_2)$	$R = x_1 * x_2 - y_1 * y_2$ $I = x_1 * y_2 + y_1 * x_2$
quoziente	$(x_1 + jy_1) / (x_2 + jy_2)$	$R = (x_1 * x_2 + y_1 * y_2) / (x_2^2 + y_2^2)$ $I = (x_2 * y_1 - x_1 * y_2) / (x_2^2 + y_2^2)$
modulo del risultato	$(R^2 + I^2)^{1/2}$	$ M = \text{SQR}(R^2 + I^2)$
argomento del risultato	$\text{Arcotang}(I / R)$	$\text{Arg}^\circ = T + 57.2957 * \text{ATN}(I / R)$

Come si evince dalle uguaglianze simboliche il Qbasic esegue le operazioni sempre e soltanto su R e su I, fornendo i risultati delle stesse nelle due parti separate. La presentazione dei vettori invece avviene come è stato mostrato del paragrafo 11.1.

11.3 Come implementare le quattro operazioni in un programma di calcolo e presentazione

La compilazione del programma di calcolo e presentazione delle quattro operazioni su due numeri complessi richiede una organizzazione delle istruzioni che è opportuno venga illustrata prima della stesura del programma stesso, essa si avvale di due nuove istruzioni **GOTO ...** ed **END**, l'istruzione (**GOTO somma**), ad esempio, che segue **IF k = ...**, invia direttamente l'esecuzione del programma alla routine di calcolo e presentazione denominata **somma**; la routine di calcolo denominata **calcom**: si ferma, dopo i calcoli finali, incontrando l'istruzione di chiusura **END**

-Le prime 15 istruzioni sono dedicate alla formazione del sistema di assi cartesiani a 4 quadranti

-La 16^a e la 17^a istruzione sono dedicate all'ingresso del coefficiente di "scala", questo valore può essere aggiustato a piacere per far sì che i vettori dei punti siano presentati al meglio nell'ambito del reticolo

-Dalla 18^a alla 21^a istruzione per l'introduzione dei valori della parte reale x_1 e della parte immaginaria y_1 relative al primo dei due numeri complessi

-Le istruzioni 22^a e 23^a permettono di stabilire quale tipo di operazione deve essere eseguita tra i due numeri complessi digitati in precedenza; con l'introduzione del valore voluto di k si potrà selezionare:

k = 1 per eseguire la somma

k = 2 per eseguire la differenza

k = 3 per eseguire il prodotto

k = 4 per eseguire il quoziente

-Dalla 24^a alla 27^a istruzione per l'introduzione dei valori della parte reale x_2 e della parte immaginaria y_2 relative al secondo dei due numeri complessi

-La 28^a istruzione per la presentazione grafica, in bianco, del vettore relativo al primo numero complesso, l'istruzione contiene i rapporti $230 / x_m$ e $160 / x_m$ che consentono di adattare al meglio la scala del reticolo alle dimensioni dei vettori

-La 29^a istruzione per la presentazione grafica, in bianco, del vettore relativo al secondo numero complesso, anche questa istruzione ha l'aggiustaggio della scala

-Dalla 30^a alla 33^a istruzione le strutture IF $k = ..$ GOTO ... che impongono al programma di saltare direttamente, in base al valore digitato di k : 1; 2; 3; 4, rispettivamente alle routine di calcolo somma; diff; prod; div

-Dalla 34^a istruzione alla 38^a la routine di calcolo per la somma dei due numeri complessi con invio alla routine comune di calcolo e presentazione dei risultati

-Dalla 39^a istruzione alla 43^a la routine di calcolo per la differenza dei due numeri complessi con invio alla routine comune di calcolo e presentazione dei risultati

-Dalla 44^a istruzione alla 48^a la routine di calcolo per il prodotto dei due numeri complessi con invio alla routine comune di calcolo e presentazione dei risultati

-Dalla 49^a istruzione alla 53^a la routine di calcolo per la divisione dei due numeri complessi con invio alla routine comune di calcolo e presentazione dei risultati

-Dalla 54^a alla 69^a istruzione per la presentazione del vettore, giallo, delle parti reale R , immaginaria I , e del calcolo e presentazione del modulo $|M|$ e dell'argomento Arg° del numero complesso risultante

Illustrato quanto sopra procediamo alla stesura del programma di calcolo e presentazione:

' FORMAZIONE DEL RETICOLO CON INDICAZIONI DI +j -j

SCREEN 9 ' (1^a istruzione)

FOR x = 0 TO 460 STEP 23

FOR y = 0 TO 320 STEP 2 ' VERTICALE

PSET (x, y), 7

NEXT y

NEXT x

LOCATE 1, 30: PRINT "+j"

LOCATE 23, 30: PRINT "-j"

FOR y = 0 TO 320 STEP 16

FOR x = 0 TO 460 STEP 3 ' ORIZZONTALE

&

```

PSET (x, y), 7
NEXT x
NEXT y
LINE ( 230, 0 )-( 230, 320 ) ' ASSE Y 4 quadranti
LINE ( 0, 160 )-( 460, 160 ) ' ASSE X 2 e 4 quadranti (15ª istruzione)
' INGRESSO DATI
LOCATE 1, 66 : PRINT "scala" ' valore di scala (16ª istruzione)
LOCATE 2, 66 : INPUT xm ' (17ª istruzione)
LOCATE 3, 66 : PRINT "x1" ' parte reale del primo numero complesso (18ª istruzione)
LOCATE 4, 66 : INPUT x1
LOCATE 5, 66 : PRINT "y1" ' parte immaginaria primo numero complesso
LOCATE 6, 66 : INPUT y1 ' (21ª istruzione)
LOCATE 7, 66 : PRINT "k" ' selezione tipo operazione k=1 (somma); k=2 (differenza);
LOCATE 8, 66 : INPUT k ' k=3 (prodotto); k=4 (divisione) (22ª e 23ª istruzione)
LOCATE 9, 66 : PRINT "x2" ' parte reale del secondo numero complesso (24ª istruzione)
LOCATE 10, 66 : INPUT x2
LOCATE 11, 66 : PRINT "y2" ' parte immaginaria del secondo numero complesso
LOCATE 12, 66 : INPUT y2 ' (27ª istruzione)
' PRESENTAZIONE DEI DUE VETTORI RELATIVI AI DUE NUMERI COMPLESSI DIGITATI
LINE ( 230, 160 )-( 230 + ( 230 / xm ) * x1, 160 - ( 160 / xm ) * y1 ) ' (28ª istruzione)
LINE ( 230, 160 )-( 230 + ( 230 / xm ) * x2, 160 - ( 160 / xm ) * y2 ) ' (29ª istruzione)
' SELEZIONE DEL TIPO DI OPERAZIONE DA ESEGUIRE
IF k = 1 GOTO somma ' (30ª istruzione)
IF k = 2 GOTO diff
IF k = 3 GOTO prod
IF k = 4 GOTO div ' (33ª istruzione)
' ROUTINE DI CALCOLO VETTORE SOMMA
somma: ' (34ª istruzione)
R = x1 + x2
I = y1 + y2
&

```

```

LOCATE 13, 60 : PRINT "SOMMA"

GOTO calcom ' (38ª istruzione)

' ROUTINE DI CALCOLO VETTORE DIFFERENZA
diff: ' (39ª istruzione)

R = x1 - x2

I = y1 - y2 '

LOCATE 13, 60 : PRINT "DIFFERENZA"

GOTO calcom ' (43ª istruzione)

' ROUTINE DI CALCOLO VETTORE PRODOTTO
prod: ' (44ª istruzione)

R = ( x1 * x2 - y1 * y2 )

I = ( x1 * y2 + y1 * x2 )

LOCATE 13, 60 : PRINT "PRODOTTO"

GOTO calcom ' (48ª istruzione)

' ROUTINE DI CALCOLO VETTORE QUOZIENTE
div: ' (49ª istruzione)

R = ( x1 * x2 + y1 * y2 ) / ( ( x2 ) ^ 2 + ( y2 ) ^ 2 )

I = ( x2 * y1 - x1 * y2 ) / ( ( x2 ) ^ 2 + ( y2 ) ^ 2 )

LOCATE 13, 60 : PRINT "QUOZIENTE"

GOTO calcom ' (53ª istruzione)

' ROUTINE DI CALCOLO COMUNE E PRESENTAZIONE
calcom: ' (54ª istruzione)

LINE( 230,160)-( 230 + ( 230/xm ) * R , 160 - ( 160/xm ) * I ),14 ' presentazione vettore risultante

M = SQR ( R^2 + I^2 ) ' calcolo del modulo vettore risultante

IF ( R = 0 ) AND ( I < > 0 ) THEN R = .00000001# ' se R=0 impone R diverso da 0; evita blocco nel calcolo Arg

IF ( R > 0 ) AND ( I = 0 ) THEN T = 0 ' sei istruzioni per individuare il quadrante nel quale

IF ( R < 0 ) AND ( I = 0 ) THEN T = 180 ' calcolare Arg°

IF ( I > 0 ) AND ( R > 0 ) THEN T = 0

IF ( I > 0 ) AND ( R < 0 ) THEN T = 180

IF ( I < 0 ) AND ( R < 0 ) THEN T = 180

```

&

```

IF (I < 0) AND (R > 0) THEN T = 360
A = T + 57.2957 * ATN ( I / R) ' calcolo Arg°
LOCATE 14, 60 : PRINT "R="; R
LOCATE 15, 60 : PRINT "I="; I
LOCATE 16, 60 : PRINT "IM="; M
LOCATE 17, 60 : PRINT "Arg°="; A
END ' (66ª istruzione)

```

Per comprendere bene l'impiego del programma che abbiamo sviluppato è necessario eseguire un esercizio per ciascun tipo di operazione ed ottenere vettori chiaramente visibili e leggibili nelle loro coordinate. Iniziamo pertanto ad eseguire la seguente somma tra numeri complessi anche se questo esercizio può essere fatto rapidamente a memoria:

$$(x_1 + jy_1) + (x_2 + jy_2) = (5 + j4) + (3 + j2)$$

dati i valori in gioco scegliamo un valore di scala 10, in questo modo ciascuna divisione del reticolo vale 1, dovendo eseguire una somma digiteremo per k il valore 1.

```

F5
                                     scala
                                     ? 10
                                     x1
                                     ? 5
                                     y1
                                     ? 4
                                     k
                                     ? 1
                                     x2
                                     ? 3
                                     y2
                                     ? 2

SOMMA
R = 8
I = 6
IM = 10
Arg° = 36.86985

```

Si ha la presentazione della parte reale R ed immaginaria I del risultato della somma, il modulo e l'argomento, nonché il tracciato dei due vettori addendi e del vettore somma così come mostrato nella figura 60.

Dalla figura si vedono chiaramente i vettori all'estremo dei quali sono collocati i punti che rappresentano i numeri complessi oggetto del nostro esercizio:

-il vettore bianco più esteso, relativo al primo numero complesso, ha l'estremo coincidente con il punto di coordinate 5; j4

-il vettore bianco meno esteso, relativo al secondo numero complesso, ha l'estremo coincidente con il punto di coordinate 3; j2

- il vettore giallo, relativo alla somma dei due numeri complessi, ha l'estremo coincidente con il punto di coordinate $R = 8; I = j6$
- l'ampiezza del modulo $|M|$ del vettore somma è pari a 10 divisioni del reticolo




Figura 60
Rappresentazione grafica della somma tra due numeri complessi

Il secondo esercizio che andiamo a svolgere è la sottrazione tra numeri complessi:

$$(x_1 + jy_1) - (x_2 + jy_2) = (-9 + j7) - (1 - j2)$$

dati i valori in gioco scegliamo un valore di scala 10, in questo modo ciascuna divisione del reticolo vale 1, dovendo eseguire una differenza digiteremo per k il valore 2.

F5

```

scala
? 10
x1
? -9
y1
? 7
k
? 2
x2
? 1
y2
? -2
DIFFERENZA
R = -10
I = 9
|M| = 13.45362
Arg = 138.0128

```

Si ha la presentazione della parte reale R ed immaginaria I del risultato della differenza, il modulo e l'argomento, nonché il tracciato dei due vettori, minuendo e sottraendo, e del vettore differenza così come mostrato nella figura 61.

Dalla figura si vedono chiaramente i vettori all'estremo dei quali sono collocati i punti che rappresentano i numeri complessi oggetto del nostro esercizio:

- il vettore bianco più esteso, relativo al primo numero complesso, ha l'estremo coincidente con il punto di coordinate $-9; j7$
- il vettore bianco meno esteso, relativo al secondo numero complesso, ha l'estremo coincidente con il punto di coordinate $1; -j2$
- il vettore giallo, relativo alla differenza dei due numeri complessi, ha l'estremo coincidente con il punto di coordinate $R = -10; I = j9$
- l'ampiezza del modulo $|M|$ del vettore differenza è pari a circa 13.4 divisioni del reticolo




Figura 61
Rappresentazione grafica della differenza tra due numeri complessi

Se ripetiamo l'esercizio di sottrazione con due numeri complessi uguali si ha:

$$(x_1 + jy_1) - (x_2 + jy_2) = (1 + j2) - (1 + j2)$$

si ottiene, naturalmente, come risultato un vettore di modulo e argomento nulli.

In questo caso l'estremo del vettore risultante coincide con l'origine degli assi e la rappresentazione grafica non è significativa.

Il terzo esercizio che sviluppiamo è relativo al prodotto tra due numeri complessi:

$$(x_1 + jy_1) \cdot (x_2 + jy_2) = (1 + j2) \cdot (3 + j4)$$

dati i valori in gioco scegliamo un valore di scala 10, in questo modo ciascuna divisione del reticolo vale 1, dovendo eseguire un prodotto digiteremo per k il valore 3.

```

F5
      scala
      ? 10
      x1
      ? 1
      y1
      ? 2
      k
      ? 3
      x2
      ? 3
      y2
      ? 4
      PRODOTTO
      R = - 5
      I = 10
      |M| = 11.18034
      Arg° = 116.5651
    
```

Si ha la presentazione della parte reale R ed immaginaria I del risultato del prodotto il modulo e l'argomento, nonché il tracciato dei due vettori, fattori, e del vettore prodotto riportati in figura 62. Dalla figura si vedono chiaramente i vettori all'estremo dei quali sono collocati i punti che rappresentano i numeri complessi oggetto del nostro esercizio:

- il vettore bianco meno esteso, relativo al primo numero complesso, ha l'estremo coincidente con il punto di coordinate 1; j2
- il vettore bianco più esteso, relativo al secondo numero complesso, ha l'estremo coincidente con il punto di coordinate 3; j4
- il vettore giallo, relativo al prodotto dei due numeri complessi, ha l'estremo coincidente con il punto di coordinate R= - 5; I = j10
- l'ampiezza del modulo |M| del vettore prodotto è pari a circa 11.2 divisioni del reticolo




Figura 62
Rappresentazione grafica
del prodotto tra
due numeri complessi

Il quarto esercizio riguarda la divisione tra due numeri complessi:

$$(x_1 + jy_1) / (x_2 + jy_2) = (-7 + j 22) / (2 + j 3)$$

Si prescinde dai valori in gioco scegliamo comunque un valore di scala 10 in favore della presentazione grafica del vettore quoziente, anche se ciò porterà fuori dal reticolo il primo numero complesso, in questo modo ciascuna divisione del reticolo vale 1, dovendo eseguire una divisione digiteremo per k il valore 4.

F5

scala
? 10
x1
? -7
y1
? 22
k
? 4
x2
? 2
y2
? 3

QUOZIENTE
R = 4
I = 5
|M| = 6.403124
Arg° = 51.34012

Si ha la presentazione della parte reale R ed immaginaria I del risultato della divisione, il modulo e l'argomento, nonché il tracciato dei due vettori, dividendo e divisore, (il vettore dividendo esce fuori dal reticolo per avere la parte immaginaria superiore alla scala assegnata) e del vettore quoziente così come mostrato nella figura 63.

Dalla figura si vedono chiaramente i vettori all'estremo dei quali sono collocati i punti che rappresentano i numeri complessi oggetto del nostro esercizio:

-il vettore bianco più esteso, relativo al primo numero complesso, ha l'estremo che esce dal reticolo dato che ha coordinate -7; j 22

-il vettore bianco meno esteso, relativo al secondo numero complesso, ha l'estremo coincidente con il punto di coordinate 2; j 3

-il vettore giallo, relativo alla divisione dei due numeri complessi, ha l'estremo coincidente con il punto di coordinate R = 4; I = j 5

-l'ampiezza del modulo |M| del vettore quoziente è pari a circa 6.4 divisioni del reticolo

11.4 Osservazioni in merito al valore di scala

Nell'ultimo esercizio del paragrafo 11.3 abbiamo visto che la scelta del valore di scala ha impedito la presentazione completa di uno dei vettori coinvolti nell'operazione di divisione. La situazione che si è verificata è stata deliberatamente voluta, sia per consentire una operazione con risultati R ed I a valori interi, sia per mostrare che a volte alcuni vettori possono uscire dal reticolo se il valore di scala non è per loro dimensionato. E' chiaro che adottando valori di scala molto più elevati dei termini che costituiscono i numeri complessi in gioco nessuno di questi uscirà dal reticolo ma si avranno magari le tracce dei vettori troppo piccole e poco visibili. Il problema può essere risolto

per il vettore risultante, facendo girare il programma almeno due volte, la prima volta con valore di scala elevato si verificano i risultati numerici in R; I. Visti questi valori si ripete la routine di programma con un valore di scala che permetta la migliore presentazione del vettore risultante dall'operazione .




Figura 63
Rappresentazione grafica
del quoziente tra due
numeri complessi

11.5 Precisazioni sulla lettura degli argomenti

Il programma che abbiamo implementato sviluppa il calcolo, tra gli altri elementi, dell'argomento del numero complesso ottenuto come risultato delle operazioni, questo valore espresso in gradi sessagesimali mediante il simbolo Arg° è stato riportato automaticamente alla fine di ciascun esercizio del paragrafo 11.3. Il valore di Arg° che misura, in senso antiorario, l'angolo formato dal vettore con l'asse dei numeri reali positivi, non è facilmente valutabile sul tracciato dato che sarebbe necessaria una misura goniometrica sullo schermo del P.C.

Per rendere concreta, almeno una volta, la corrispondenza tra il valore dell'argomento scaturito dal calcolo e il valore dell'argomento visibile sul tracciato è utile un semplice esercizio di differenza tra due numeri complessi che ha come risultato un numero complesso dotato di argomento $\text{Arg}^\circ = 45$, in questo caso vedremo che il vettore risultante biseca esattamente il primo quadrante denunciando un argomento pari a $(90 / 2)^\circ$.

Con l'ausilio del programma del paragrafo 11.3 eseguiamo l'operazione:

$$(x1 + jy1) - (x2 + jy2) = (6 + j5) - (2 + j1)$$

F5

scala
? 10
x1
? 6
y1

z_2
 k
 z_1
 x_2
 y_2
 x_1
 y_1
 DIFFERENZA
 $R = 4$
 $I = 4$
 $|M| = 5.656854$
 $\text{Arg}^\circ = 44.99994$




Figura 64
 Rappresentazione grafica
 della differenza tra
 due numeri complessi
 Vettore risultante $\text{Arg}^\circ = 45$

Otteniamo il grafico di figura 64 dove si vedono chiaramente i vettori all'estremo dei quali sono collocati i punti che rappresentano i numeri complessi oggetto del nostro esercizio, in particolare il vettore giallo, relativo al risultato della sottrazione dei due numeri complessi, ha l'estremo coincidente con il punto di coordinate $R = 4$; $I = j 4$ e taglia in due il primo quadrante formando un angolo di 45° così come espresso numericamente dal calcolo di Arg° mediante la presentazione del valore $\text{Arg}^\circ = 44.99994$.

11.6 I numeri complessi funzione di un'unica variabile reale

Sovente si incontrano numeri complessi, le cui parti reale, ed immaginaria, sono in funzione di una variabile indipendente che ne condiziona il posizionamento nel sistema di assi cartesiani.

Infatti se un numero complesso

$$N = x_1 + j y_1$$

è legato ad s tramite le funzioni

$$x_1 = f_1(s); \quad y_1 = f_2(s)$$

ad ogni valore di s corrisponderà un numero complesso e, a questo, un punto rappresentativo come estremo del vettore che lo caratterizza.

Le due parti di N possono ad esempio dipendere da s secondo le funzioni:

$$x1 = (s^3 / 30) + .1$$

$$y1 = (\text{Sen } s)^3 + .2$$

ci saranno pertanto, per ogni valore assegnato alla variabile indipendente s, una coppia di valori x1 ed y1 che individueranno un ben determinato numero complesso $x1+j y1$; per $s = 1.5$ radianti, ad esempio, si ha

$$x1 = (1.5^3 / 30) + .1 = .2125$$

$$y1 = (\text{Sen } 1.5)^3 + .2 = 1.1925$$

ed N assume il valore: $N = .2125 + j 1.1925$

Se ora vogliamo rappresentare graficamente una serie di 10 numeri complessi generati dalle funzioni sopra espresse per la s variabile da -3 radianti a + 3 radianti dobbiamo ricorrere al programma illustrato nel paragrafo 11.1 apportando ad esso alcune modifiche, sia per calcolare per ciascun valore di s la corrispondente coppia x1;y1, sia per evitare di dover digitare 10 istruzioni del tipo LINE per il tracciamento dei 10 vettori calcolati. Per queste modifiche dobbiamo anzitutto trasformare in Qbasic le funzioni date:

simbologia ordinaria

$$x1 = (s^3 / 30) + .1$$

$$y1 = (\text{Sen } s)^3 + .2$$

simbologia Qbasic

$$x1 = ((s^3) / 30) + .1$$

$$y1 = ((\text{SIN}(s))^3) + .2$$

e di seguito compilare il programma commentato con le modifiche e gli adattamenti di scala necessari:

```
' FORMAZIONE DEL SISTEMA DI ASSI CARTESIANI A 4 QUADRANTI CON INDICAZIONI
```

```
' SULL'ASSE DELLE ORDINATE DEI SIMBOLI +j ; -j
```

```
SCREEN 9
```

```
FOR x = 0 TO 460 STEP 23
```

```
FOR y = 0 TO 320 STEP 2 ' VERTICALE
```

```
PSET (x, y), 7
```

```
NEXT y
```

```
NEXT x
```

&

```

LOCATE 1, 30: PRINT "+j" ' presentazione del simbolo +j
LOCATE 23, 30: PRINT "-j" ' presentazione del simbolo -j
FOR y = 0 TO 320 STEP 16
FOR x = 0 TO 460 STEP 3 ' ORIZZONTALE
PSET (x, y), 7
NEXT x
NEXT y
LINE (230, 0)-(230, 320) ' ASSE Y 4 quadranti
LINE (0, 160)-(460, 160) ' ASSE X 4 quadranti
' CALCOLO DEI VALORI DI x1 e y1 IN FUNZIONE DI s
FOR s = -3 TO 3 STEP .6 ' stabilisce il campo e l'incremento della variabile indipendente s
x1 = ((s ^ 3) / 30) + .1 ' calcola la parte reale del numero complesso funzione di s
y1 = ((SIN(s)) ^ 3) + .2 ' calcola la parte immaginaria del numero complesso funzione di s
' COMPOSIZIONE ISTRUZIONE LINE SULLA BASE DEI VALORI CALCOLATI DI x1 E y1
LINE (230,160) - (230 + 230 * x1, 160 - 130 * y1),14 ' traccia i vettori calcolati in giallo
NEXT s ' rimanda all'istruzione FOR s=... per il calcolo dei successivi numeri complessi

```

Come si può vedere il programma è molto semplice, non resta che provarlo: F5
Viene presentato il grafico di figura 65 in cui compaiono, sul reticolo, un ventaglio di 10 vettori.




Figura 65
Rappresentazione grafica
di numeri complessi funzione
di variabile reale.
metodo dei vettori

La presentazione grafica dei numeri complessi che abbiamo ora mostrato è di importanza fondamentale nello studio di alcuni fenomeni fisici, essa permette di inquadrare l'andamento complessivo dei vettori, funzioni della variabile indipendente s , sia nella sequenza di rotazione dell'argomento, sia nella variazione del modulo. Questo tipo di grafica non si adatta bene nel caso in cui i vettori ruotino tanto da intersecare altri vettori, in tal caso si deve ricorrere ad una soluzione diversa.

La soluzione consiste nell'adottare l'istruzione PSET invece dell'istruzione LINE, rinunciando alla grafica dei vettori in favore della grafica dei punti nel piano complesso.

Se ripetiamo l'esempio precedente con l'istruzione PSET abbiamo il seguente programma modificato che pone in evidenza soltanto gli estremi dei vettori collocati nell'ambito del sistema di assi cartesiani a 4 quadranti.

SCREEN 9

FOR x = 0 TO 460 STEP 23

FOR y = 0 TO 320 STEP 2 ' VERTICALE

PSET (x, y), 7

NEXT y

NEXT x

LOCATE 1, 30: PRINT "+j" ' presentazione del simbolo +j

LOCATE 23, 30: PRINT "-j" ' presentazione del simbolo -j

FOR y = 0 TO 320 STEP 16

FOR x = 0 TO 460 STEP 3 ' ORIZZONTALE

PSET (x, y), 7

NEXT x

NEXT y

LINE (230, 0)-(230, 320) ' ASSE Y 4 quadranti

LINE (0, 160)-(460, 160) ' ASSE X 4 quadranti

' CALCOLO DEI VALORI DI x_1 e y_1 IN FUNZIONE DI s

FOR s = -3 TO 3 STEP .6 ' stabilisce il campo e l'incremento della variabile indipendente s

$x_1 = ((s^3)/3) + .1$ ' calcola la parte reale del numero complesso funzione di s

$y_1 = ((\sin(s))^3) + .2$ ' calcola la parte immaginaria del numero complesso funzione di s

' COMPOSIZIONE ISTRUZIONE PSET SULLA BASE DEI VALORI CALCOLATI DI x_1 E y_1

PSET (230 + 230 * x_1 , 160 - 130 * y_1),14 ' insieme di 5 istruzioni per tracciare i punti

PSET (230 + 230 * x_1 + 1, 160 - 130 * y_1),14 ' calcolati mediante crocette

PSET (230 + 230 * x_1 - 1, 160 - 130 * y_1),14

&

```
PSET ( 230 + 230 * x1 , 160 - 130 * y1 + 1 ),14
```

```
PSET ( 230 + 230 * x1 , 160 - 130 * y1 - 1 ),14
```

```
NEXT s ' rimanda all'istruzione FOR s=... per il calcolo dei successivi numeri complessi
```

F5

A seguito dell'azione su F5 il programma rende visibili i 10 punti oggetto del calcolo.

La nuova grafica di presentazione è mostrata in figura 66.

Dalla figura si vede che compaiono soltanto una sequenza di punti che descrivono le posizioni dei 10 numeri nel piano complesso, questi sono gli estremi dei vettori i cui moduli ora non sono visualizzati. Vedremo in seguito come questo tipo di grafica sia indispensabile per visualizzare particolari serie di numeri complessi. E' superfluo sottolineare che i punti che compaiono nel piano complesso **non sono** funzioni dei valori dell'asse delle ascisse, come nel caso delle funzioni di tabella, ma sono legati soltanto dalla variabile indipendente s che non compare nel grafico in alcun modo.




Figura 66
Rappresentazione grafica
di numeri complessi
funzione di variabile reale
metodo dei punti

11.7 La grafica dei numeri complessi di un sistema risonante

L'impiego della grafica Qbasic ha una delle sue più interessanti applicazioni nella visualizzazione dei numeri complessi che esprimono l'andamento di un fenomeno fisico relativo ai sistemi risonanti in generale. Del fenomeno in questione mostriamo soltanto l'algoritmo complesso che ne rappresenta l'ammettenza, senza entrare nel merito delle problematiche teoriche e concettuali che lo riguardano. Il numero complesso in oggetto può essere scritto nel modo solito:

$$N = x1 + jy1$$

mentre la dipendenza delle parti reale ed immaginaria dalla variabile indipendente s è data da:

$$x1 = \frac{1}{1 + (s - 1/s)^2} + 1$$

$$y1 = s - \frac{s - 1/s}{1 + (s - 1/s)^2}$$

Volendo eseguire come esercizio la rappresentazione grafica di questo numero complesso, al variare di s , le due funzioni devono essere trasformate in Qbasic per l'implementazione nel programma, si ha perciò:

$$x1 = (1 / (1 + (s - 1/s)^2)) + 1$$

$$y1 = s - ((s - 1/s) / (1 + (s - 1/s)^2))$$

La compilazione del programma deve essere preceduta da alcune definizioni numeriche:

Per rendere perspicuo l'esercizio si sceglie un campo di variabilità della s compreso tra $s = .1$ e $s = 2.5$ con passi di $.1$ in modo da calcolare 25 numeri complessi.

Si fissa per l'istruzione PSET: $k = 230 / 4 = 57.5$ e $k2 = 160 / 4 = 40$

Con questi valori e con le funzioni trasformate in Qbasic si compila il programma:

SCREEN 9

FOR x = 0 TO 460 STEP 23

FOR y = 0 TO 320 STEP 2 ' VERTICALE

PSET (x, y), 7

NEXT y

NEXT x

LOCATE 1, 30: PRINT "+j" ' presentazione del simbolo +j

LOCATE 23, 30: PRINT "-j" ' presentazione del simbolo -j

FOR y = 0 TO 320 STEP 16

FOR x = 0 TO 460 STEP 3 ' ORIZZONTALE

PSET (x, y), 7

NEXT x

&

```

NEXT y
LINE ( 230, 0)-( 230, 320 ) ' ASSE Y 4 quadranti
LINE ( 0, 160)-( 460, 160 ) ' ASSE X 4 quadranti
' CALCOLO DEI VALORI DI x1 e y1 IN FUNZIONE DI s
FOR s = .1 TO 2.5 STEP .1 ' stabilisce il campo e l'incremento della variabile indipendente s
x1 = ( 1 / ( 1 + ( s - 1/s ) ^ 2 ) ) + 1 ' calcolo della parte reale funzione di s
y1 = s - ( ( s - 1/s ) / ( 1 + ( s - 1/s ) ^ 2 ) ) ' calcolo della parte immaginaria funzione di s
' COMPOSIZIONE ISTRUZIONE PSET SULLA BASE DEI VALORI CALCOLATI DI x1 E y1
PSET ( 230 + 57.5 * x1 , 160 - 40 * y1 ),14 ' insieme di 5 istruzioni
PSET ( 230 + 57.5 * x1 + 1 , 160 - 40 * y1 ),14 ' per tracciare i punti calcolata mediante
PSET ( 230 + 57.5 * x1 - 1 , 160 - 40 * y1 ),14 ' crocette
PSET ( 230 + 57.5 * x1 , 160 - 40 * y1 + 1 ),14
PSET ( 230 + 57.5 * x1 , 160 - 40 * y1 - 1 ),14
NEXT s ' rimanda all'istruzione FOR s=... per il calcolo dei successivi numeri complessi

```

F5

L'esecuzione del programma porta al grafico di figura 67.




Figura 67
Rappresentazione grafica
di numeri complessi
funzione di variabile reale
ammettenza di un sistema
risonante

Il grafico ottenuto mostra la notevole capacità espressiva del mezzo di presentazione; si osserva, tracciato nel primo quadrante, l'insieme dei 25 punti disposti secondo un pseudo nodo che mette in evidenza, all'occhio esperto, la caratteristica di risonanza del sistema fisico.

Da questo esercizio si comprende bene che se avessimo tracciato i 25 vettori questi avrebbero completamente coperto il nodo cancellando di fatto le informazioni che invece sono state fornite.

Si conclude che il sistema grafico deve essere adattato, con un poco di esperienza, al tipo di informazioni che si desiderano dal calcolo.

Si deve inoltre osservare che aver limitato a 25 i punti sul piano consente di associare, a vista, a ciascuno di essi il valore della variabile indipendente che lo ha generato; infatti il primo punto verso l'asse delle ascisse rappresenta il numero complesso ottenuto per $s = .1$, il punto successivo, verso l'alto, rappresenta il numero complesso ottenuto per $s = .2$, e così via fino al punto più in alto generato dal valore di $s = 2.5$. La possibilità di questa importantissima correlazione visiva, tra valore della variabile indipendente s e posizione dei punti nel piano complesso, è una ulteriore fonte di informazione che non si potrebbe avere se il numero dei punti fosse talmente elevato da fornire una curva continua invece che una netta punteggiata.

Si può pertanto concludere che il sistema grafico, se correttamente utilizzato, è in grado di fornire con immediatezza una quantità di informazioni riguardanti il sistema risonante, che non sarebbero facilmente acquisibili tramite un tabulato numerico che raccogliesse i 25 numeri complessi, sia che fossero espressi in forma cartesiana, sia che fossero espressi in forma polare.

11.8 La grafica complementare dei numeri complessi di un sistema risonante

Per completare l'esposizione delle metodologie grafiche e numeriche relative all'impiego dei numeri complessi per lo studio di un sistema risonante è necessario accennare alla grafica complementare che rappresenta un utile corollario all'esposizione fatta nel paragrafo 11.7.

Per ricavare ulteriori informazioni dalla serie dei numeri complessi che rappresentano un fenomeno fisico sono disponibili altre versioni grafiche delle stesse per evidenziarne:

- andamento della parte reale in funzione di s
- andamento della parte immaginaria in funzione di s
- andamento del modulo in funzione di s
- andamento dell'argomento in funzione di s

Queste funzioni di s , che caratterizzano la serie dei numeri complessi, sono funzioni nel campo reale e possono essere rappresentate con un numero elevato di punti dato che in questo caso la variabile indipendente si trova collocata sulle ascisse e il riferimento con i punti tracciati è immediato.

Per completare l'esercizio del paragrafo 11.7 impostiamo il nuovo programma per il tracciamento della parte reale x_1 come segue:

Dal grafico di figura 67 si rileva che il valore più elevato della parte reale si ha all'estrema destra del nodo per $s = 1$ a cui corrisponde $x_1 = 2$, questo è il valore sul quale deve essere adattata la scala delle ordinate.

Per la scala delle ascisse il campo di s dovrà avere gli stessi valori imposti nell'esercizio sopra citato: da $s = .1$ a $s = 2.5$.

La figura 67 mostra inoltre che il campo di variabilità di x_1 è sempre positivo, si adatta perciò una presentazione ad un solo quadrante, per la quale l'istruzione PSET diventa:

$$\text{PSET}(k_1 * s, 320 - k_3 * x_1) \quad \text{dove}$$
$$k_1 = 460 / s = 460 / 2.5 = 184$$
$$k_3 = 320 / x_1 = 320 / 2 = 160$$

Ciò premesso abbiamo:

```
LINE ( 0 , 320 ) - ( 460 , 320 ) ' comanda il tracciamento dell'asse delle ascisse (asse s)
' per coordinate ad I quadrante

LINE ( 0 , 0 ) - ( 0 , 320 ) ' comanda il tracciamento dell'asse delle ordinate (asse x1)
' per coordinate ad I quadrante

LOCATE 1,66 : PRINT " PARTE REALE" ' indicazione della funzione presentata

FOR s = .1 TO 2.5 STEP .001 ' campo di variabilità ed incremento della s per avere 2500 punti di calcolo

x1 = ( 1 / ( 1 + ( s - 1/s ) ^ 2 ) ) + 1 ' calcolo della parte reale funzione di s

PSET( 184 * s , 320 - 160 * x1 ),14 ' presentazione della parte reale come una curva continua

NEXT s ' comanda il programma al ritorno automatico all'istruzione FOR s = ecc.
```

F5

si ha la presentazione del grafico dell'andamento della parte reale in funzione di s così come mostrato in figura 68.

La figura evidenzia molto chiaramente come varia la parte reale del numero complesso al variare di s .

La funzione cresce, con l'incremento di s , ed ha il suo massimo per $s = 1$ dopo di che decresce con regolarità.

In questo caso la curva è continua e il legame tra x_1 ed s è diretto; si può pertanto leggere, con l'ausilio del reticolo, che valore assume la parte reale del numero complesso per un qualsiasi valore di s compreso nel campo di variabilità assegnato.

Informazioni di questo tipo, con la precisione offerta dalla nuova grafica, non sono certamente ricavabili dalla figura 67 che ha invece lo scopo preciso di fornire un andamento globale della serie dei numeri complessi.




Figura 68
Rappresentazione grafica
della parte reale
dell'ammettenza
del sistema risonante

Analogamente alla parte reale si traccia l'andamento della parte immaginaria relativa alla serie dei numeri complessi sulla base dei seguenti dati:

Dal grafico di figura 67 si rileva che il valore più elevato della parte immaginaria si ha all'estremo superiore della punteggiata per $s = 2.5$ a cui corrisponde $y_1 = 2.1$ che per facilitare la lettura del grafico si porta a 2.5, questo è il valore sul quale deve essere adattata la scala delle ordinate.

Per la scala delle ascisse il campo di s dovrà avere gli stessi valori imposti in precedenza: da $s = .1$ a $s = 2.5$.

La figura 67 mostra inoltre che il campo di variabilità di y_1 è sempre positivo, si adatta perciò una presentazione ad un solo quadrante, per la quale l'istruzione PSET diventa

PSET($k_1 * s$, $320 - k_3 * y_1$) dove $k_1 = 460 / s = 460 / 2.5 = 184$
 $k_3 = 320 / y_1 = 320 / 2.5 = 128$

Il programma per la presentazione della parte immaginaria sarà pertanto:

```
LINE ( 0 , 320 ) - ( 460 , 320 ) ' comanda il tracciamento dell'asse delle ascisse (asse s)
                                ' per coordinate ad I quadrante

LINE ( 0 , 0 ) - ( 0 , 320 ) ' comanda il tracciamento dell'asse delle ordinate (asse y1)
                                ' per coordinate ad I quadrante

LOCATE 1,66: PRINT "PARTE IMMAG." ' indicazione della funzione presentata

FOR s = .1 TO 2.5 STEP .001 ' campo di variabilità ed incremento della s per avere 2500 punti di calcolo

y1 = s - ((s - 1/s) / (1 + (s - 1/s)^2)) ' calcolo della parte immaginaria funzione di s

PSET( 184 * s , 320 - 128 * y1 ),14 ' presentazione della parte immaginaria come una curva continua

NEXT s ' comanda il programma al ritorno automatico all'istruzione FOR s = ecc.
```

Premendo F5 si ha la presentazione del grafico dell'andamento della parte immaginaria in funzione di s così come mostrato in figura 69.




Figura 69
 Rappresentazione grafica
 della parte immaginaria
 dell'ammettenza del
 sistema risonante

La figura evidenzia come varia la parte immaginaria del numero complesso al variare di s . Anche in questo caso la curva è continua e il legame tra y_1 ed s è diretto, si può pertanto leggere, con l'ausilio del reticolo, che valore assume la parte immaginaria del numero complesso per un qualsiasi valore di s compreso nel campo di variabilità assegnato. La funzione ha un punto di massimo per $s = .75$ ed un punto di minimo per $s = 1.2$.

Per la presentazione grafica del modulo della serie di numeri complessi si deve iniziare, come per le altre parti, dall'esame del grafico globale di figura 67 in cui risulta che la distanza massima tra l'origine degli assi e la punteggiata si ha per il punto corrispondente al valore di $s = 2.5$; sulla base di questo numero si deve calcolare il massimo valore del modulo con il programma:

CLS

INPUT " s=" ; s

x1 = (1 / (1 + (s - 1/s) ^ 2)) + 1

y1 = s - ((s - 1/s) / (1 + (s - 1/s) ^ 2))

M = SQR(x1 ^ 2 + y1 ^ 2)

PRINT "IM=" ; M

che per $s = 2.5$ dà $IM = 2.4$ che si arrotonda per comodità di lettura delle scale in 2.5.

Per la scala delle ascisse il campo di s dovrà avere gli stessi valori imposti in precedenza:

da $s = .1$ a $s = 2.5$.

Dato che il modulo di un numero complesso è sempre positivo, si adotta anche in questo caso una presentazione ad un solo quadrante, per la quale l'istruzione PSET diventa

PSET(k1 * s , 320 - k3 * x1) dove $k1 = 460 / s = 460 / 2.5 = 184$
 $k3 = 320 / y1 = 320 / 2.5 = 128$

Il programma per la presentazione dell'andamento del modulo è:

LINE (0 , 320) - (460 , 320) ' comanda il tracciamento dell'asse delle ascisse (asse s)
 ' per coordinate ad 1 quadrante

LINE (0 , 0) - (0 , 320) ' comanda il tracciamento dell'asse delle ordinate (asse IM)
 ' per coordinate ad 1 quadrante

LOCATE 1,66 : PRINT "MODULO" ' indicazione della funzione presentata

FOR s = .1 TO 2.5 STEP .001 ' campo di variabilità ed incremento della s per avere 2500 punti di calcolo

x1 = (1 / (1 + (s - 1/s) ^ 2)) + 1 ' calcolo della parte reale

y1 = s - ((s - 1/s) / (1 + (s - 1/s) ^ 2)) ' calcolo della parte immaginaria

M = SQR(x1 ^ 2 + y1 ^ 2) ' calcolo del modulo

PSET(184 * s , 320 - 128 * M),14 ' presentazione del modulo come una curva continua

NEXT s ' comanda il programma al ritorno automatico all'istruzione FOR s = ecc.

F5

si ha la presentazione del grafico dell'andamento del modulo in funzione di s così come mostrato in figura 70. Il modulo ha un massimo per $s = .94$, un minimo per $s = 1.6$.




Figura 70
Rappresentazione grafica
del modulo dell'ammettenza
del sistema risonante

Vediamo infine la presentazione degli argomenti della serie di numeri complessi che deve iniziare dall'esame del grafico globale di figura 67 in cui risulta che tutti i punti sono contenuti nel primo quadrante, ciò dà modo di affermare che gli argomenti stessi sono tutti contenuti entro 90° , si ha perciò:

Per la scala delle ascisse il campo di s dovrà avere gli stessi valori imposti in precedenza: da $s = .1$ a $s = 2.5$.

Dato che l'argomento è al massimo 90° , si adotta anche in questo caso una presentazione ad un solo quadrante, per la quale l'istruzione PSET diventa:

PSET($k1 * s, 320 - k3 * x1$) dove $k1 = 460 / s = 460 / 2.5 = 184$
 $k3 = 320 / \text{Arg}^\circ = 320 / 90 = 3.55$

Il programma per la presentazione dell'andamento dell'argomento è:

```
LINE ( 0 , 320 ) - ( 460 , 320 ) ' comanda il tracciamento dell'asse delle ascisse (asse s)
                               ' per coordinate ad 1 quadrante

LINE ( 0 , 0 ) - ( 0 , 320 )   ' comanda il tracciamento dell'asse delle ordinate (asse Arg°)
                               ' per coordinate ad 1 quadrante

LOCATE 1,66 : PRINT "ARGOMENTO" ' indicazione della funzione presentata

FOR s = .1 TO 2.5 STEP .001 ' campo di variabilità ed incremento della s per avere 2500 punti di calcolo

x1 = ( 1 / ( 1 + ( s - 1/s ) ^ 2 ) ) + 1 ' calcolo della parte reale

y1 = s - ( ( s - 1/s ) / ( 1 + ( s - 1/s ) ^ 2 ) ) ' calcolo della parte immaginaria
```

$A = 57.2957 * \text{ATN}(y1 / x1)$ ' calcolo dell'argomento in gradi sessagesimali

$\text{PSET}(184 * s, 320 - 3.55 * A), 14$ ' presentazione dell'argomento come una curva continua

$\text{NEXT } s$ ' comanda il programma al ritorno automatico all'istruzione $\text{FOR } s =$ ecc.

F5

si ha la presentazione del grafico dell'andamento dell'argomento in funzione di s così come mostrato in figura 71. L'argomento ha un massimo per $s = .56$ ed un minimo per $s = 1.2$.




Figura 71
Rappresentazione grafica
argomento di numeri complessi
funzione di variabile reale